
 I. INTRODUCTION

The distributed nature of many microwave and millimeter-wave circuits necessitates
electromagnetic modeling. Thus the full application of computer-aided engineering (CAE) to these
circuits requires the integration of electromagnetic models of distributed structures with
conventional circuit analysis. In the formative stages of the CAE of microwave circuits, ports were
used in specifying the interconnectivity of networks.  The utilization of ports avoided the issue of
specifying reference nodes.  In analysis, using matrix manipulations or signal flow graphs, one of
the terminals of a port was used implicitly as a reference node and generally ignored in formulating
the mathematical model.  In one commercial nonlinear microwave simulator ports are still used in
circuit specification.  The connection of discrete elements is specified nodally but at the highest
level of the hierarchy port-based descriptions are used. While it is possible to analyze any circuit
with this arrangement it becomes increasingly difficult to specify the connections of large spatially
distributed circuits, and also to specify and extract desired output quantities.  The alternative to
using a port-only descriptions is to exclusively use the nodal connectivity description – the only
method used in general purpose circuit simulators.  The conventional nodal specification enables
circuit elements to be connected in any possible combination and only one reference node
(commonly called the global reference node or simply ground) is used.  So nodal specification
provides tremendous flexibility in circuit specification but the multiple reference nodes leads to M-
fold indefiniteness of the (modified) nodal admittance matrix.  Also,  with spatially distributed
circuits it is possible to make illegal connections such as connecting a non-spatially distributed
element, say a resistor, across a spatially distributed element, e.g. a transmission line.  The purpose
of this paper is two-fold: firstly to describe an implementation of the local reference node theory
described by Khalil and Steer in Reference [7]. Secondly to develop and describe the
implementation of a scheme for checking the legal connections of a network with mixed spatially
distributed and conventional elements.



2

II. BACKGROUND

A circuit is a graphical construct (which can be specified in textual form as a netlist) for coupling
together algebraic and first order differential equations. These equations arise from the constitutive
relations of the individual elements, which specify the actual form of the individual equations, and
from Kirchoff’s current law which specifies how the equations are coupled. With but a few
modifications, the analysis thus described is called nodal admittance analysis. It is important to
note that circuits have no sense of space – a circuit is defined as though it existed at a point. The
problem is then how to incorporate an electromagnetic model of a structure that is inherently
distributed in space. Of two approaches, one is to insert the device equations into an appropriate
time-stepping electromagnetic simulator such as a finite difference time domain (FDTD) simulator
[1, 2]. This reduces the level of abstraction of the “circuit” and embeds the constitutive relations of
the conventional circuit elements into the analysis grid of the FDTD method. An alternative is to
retain the high-level circuit abstraction and incorporate the results of a field analysis (the spatially
distributed circuit) into a circuit structure [3, 4, 5, 6]. Thus the problem is one of taking an
electromagnetic solution and using a physically consistent approach to inserting it is a circuit
element. This is a less direct approach, that is more abstract, but potentially much more general and
compatible with the large body of existing computer-aided circuit analysis theory and knowledge
that has been developed.

At this point it is useful to review modified nodal admittance (MNA) analysis to indicate the basis
for the work presented here. MNA analysis was developed to handle elements that do not have
nodal admittance descriptions.  For each such element one or more additional equations are added
to the nodal admittance equations and these new equations become additional rows and columns in
the evolving matrix system of equations.  A similar approach can be followed for the
electromagnetic elements.  The process is a little more sophisticated as it is no longer sufficient to
add additional rows.  Instead the concept of local reference nodes was developed [7] as a
generalization of the compression matrix approach [6]. This concept provides another way to
incorporate alternate equations in the evolving MNA matrix. However, rather than adding
additional constitutive relations, the local reference node concept changes the way the port-based
parameters are used. Figure 2 shows a circuit with a spatially distributed circuit and with local
reference nodes indicated by the diagonal symbol.  In a conventional circuit only one reference
node (ground) is possible so that application of KCL to the global reference node introduces just
one additional redundant row and column in the indefinite form of the MNA matrix. For a spatially
distributed circuit, KCL is applied to each locally referenced group one at a time and, as the local
reference nodes are electrically connected only through a spatially distributed element, each
application results in a redundant row and column in the MNA matrix. This is a particular property
of the port-based characterization of the spatially distributed element [7]. Thus only appropriate
electromagnetic analyses are suitable. The mathematical description of the required extension to
the nodal admittance analysis is given in [7].

III. METHOD
The basis of the method is that in a multi-port element, the terminals of ports with different

local reference nodes can be considered isolated.  For example, if we write the equivalent circuit of
a two-element derived from the port-based admittance (y) parameters,

we obtain the equivalent circuit in Figure 1. Note that no current can flow between the two ports.
This type of circuit model can be generalized for an n-port element where external and internal
local reference nodes are defined. The local references shown in Figure 1 are internal because they
are used internally in the element to measure the voltages at its ports. An external local reference
terminal, on the other hand, is an arbitrarily chosen terminal from a locally referenced group. In
general, for any spatially distributed circuit (Figure 2) there is no current between port groups
inside the spatially-distributed element. Thus the circuit can be divided into subcircuits, each with a
local reference terminal, as shown in Figure 3. Each subcircuit is isolated with respect to the others,
so there is no change in the solution if all the reference nodes are connected together. The
difference is that now there is only one global reference node for all the circuit, and standard nodal
methods can be used to formulate the circuit equations. The problem becomes that of detecting
violations to the assumption that each locally referenced group is isolated from the others and that
there is exactly one reference terminal for each one. Such a situation is shown in Figure 4. The
effect of this connection is the creation of a current loop that is non-physical since the two circuits
cannot be connected instantaneously. On the other hand, it is valid for two subcircuits to be
connected by more than one spatially distributed element. It is also possible for two ports of a
spatially distributed element to be connected to the same locally referenced group (for example a
delay line). In that case, the external local reference node would be the same for both ports of the
line, but internally, the line still has two local reference nodes.

After all the checking is done, all the local reference nodes are merged into a single global
reference node, and the solution of the circuit is found by standard procedures. The nodal voltages
found are coincident with the voltages referred to each local reference node.

IV. IMPLEMENTATION

Implementation of the technique can be described using a conventional procedural approach or an
object oriented approach.  While not obvious to those not conversant with object oriented practice,
the object oriented view maps onto the circuit analysis problem cleanly.  A circuit is a collection of
objects (resistors, inductors, transmission lines, etc.) that are related to each other (they share
nodes).  With the introduction of spatially distributed elements along with the conventional
elements there are certain rules that describe allowable interconnections. The rule is

Spatially distributed objects can be connected together in any manner but the
interconnections of non-spatially distributed elements cannot span spatially
distributed elements.

In Transim (our simulator), each element and each terminal is a node in a graph structure. The
algorithm to detect violations is a variation of the depth-search algorithm and is best illustrated
using what is called a collaboration diagram. The algorithm could also be implemented using
conventional procedural programming [8]. The collaboration diagram [9] is part of the Unified
Modeling Language (UML), a language for specifying, visualizing, and constructing the artifacts
of software systems as well as for business modeling. The UML represents a collection of “best
engineering practices” that have proven successful in the modeling of large and complex systems.
Behavior is implemented by sets of objects that exchange messages within an overall interaction to
accomplish a purpose. To understand the mechanisms used in a design, it is important to see only
the objects and the messages involved in accomplishing a purpose or a related set of purposes,
projected from the larger system of which they are part for other purposes. Such a static construct
is called a collaboration. A collaboration is a set of participants and relationships that are
meaningful for a given set of purposes. The identification of participants and their relationships
does not have global meaning. In the collaboration diagram, each box represents an object (in this
case, either a terminal or an element). The lines between boxes represent associations and the
arrows are messages. The order in which messages are sent is shown by the numbers.

Figure 5 shows the circuit view corresponding to the collaboration diagram of Figure 6. The
terminal numbers indicate the order in which terminals are discovered by the algorithm. Note also
that the internal local reference terminals of the spatially distributed element do not need to be
coincident with the external references. The search begins at each local reference terminal. The
local reference terminal sends a message to propagate its identifier (id) to all the adjacent elements.
If the element that receive the message has only one local reference terminal (lumped), it continues
the propagation to the other adjacent terminals. Otherwise it only propagates the id to the element
terminals with the same internal reference. If an element attempts to propagate a reference id to a
terminal already marked with another reference id, then a violation is detected: there is an illegal
connection between two locally referenced groups. A check to detect floating parts of the circuit
can be made by finding any terminal that does not belong to any locally referenced group.  All
these checks are implemented in Transim. Note that the bias circuit often violates the condition that
the locally referenced groups are isolated between them. To overcome this either the distributed
elements should include the DC characteristics or an independent DC source should be considered
in each group.   In the input netlist, the local reference nodes are identified with the command
“ .ref < terminal >”. By default the nodes named “0” and “gnd ” are assumed to be reference
nodes.

V DISCUSSION

The concepts and algorithms presented in this paper are well illustrated using a spatially distributed
spatial power combining circuit [10, 11] as an example. The system is shown in Figure 7 where a
signal transmitted by a horn antenna illuminates a 2x2 grid of amplifier unit cells. The planar array
of amplifiers is shown in greater detail in Figure 8 where each unit cell is localized at a grid
intersection and comprises two MMIC’s in a push-pull configuration. Each unit cell is separated
from its neighbors and each radiates amplified power at the orthogonal polarization to the input
signal.  The contributions from each cell combine spatially and the combined power is collected by
the output horn antenna.  The electromagnetic environment is modeled using Method of Moments
(MOM) techniques [10].  Clearly a global reference node cannot be defined for this electrically
large structure.  Instead local reference nodes are defined, one per unit cell and one for each of the

bias circuit connections, so that there are sixteen local reference nodes.  The Spice-like netlist for
this system is given in Figure 9. In this netlist there are three special element types, namely the
ymatrix, gridex and vct.  The ymatrix element is used to define a multi-port element described by
its y parameters.  The information to build the element is stored in a file that contains the y
parameters as well as the port information. In this example, the element was used to model the
quasi-optical grid and the element file was generated using an electromagnetic simulator called
yomoma. The gridex is an element to provide the field excitation to the grid. From the circuit point
of view it looks like a set of current sources.  The input files were also generated by the yomoma
MOM program [10].  The vct element is a nonlinear voltage-controlled current source.  It is used
inside the MMIC behavioral model. Note that the y matrix defined element generated by yomoma
set to zero the DC matrix. Therefore the bias circuit can not be modeled inside the spatially
distributed element and must be implemented using standard lumped elements. This could lead to a
violation of the condition that the locally referenced groups are isolated between them. To
overcome this, either the distributed elements should include the DC characteristics or an
independent bias circuit must be used in each locally referenced group.

VI CONCLUSION

The approach described here enables conventional circuit simulators to be extended to incorporate
spatially distributed circuits.  The implementation is an extension of modified nodal admittance
analysis and the overhead involved in implementing the scheme is small. The authors have used the
work described here for the steady-state modeling of quasi-optical power combining circuits [10,
11].  In conventional nodal-based analysis the row and column associated with the global reference
node are not inserted into the evolving modified nodal admittance matrix.  The same inspection
procedure is used with locally referenced nodes.  Thus the additional row and column associated
with each locally referenced node are not incorporated into the evolving matrix equation and so
avoiding the substantial performance penalty that would be incurred with explicit row and column
elimination.  If required these rows and columns can always be reconstructed by knowing the
members of a locally referenced group [7].   The effective elimination of the one row and column
per locally referenced leads to a non-singular matrix.

CAPTIONS

Figure 1. Equivalent circuit of a two-port distributed element.

Figure 2  Generic spatially distributed circuit.

Figure 3. The locally referenced groups are isolated, therefore the solution is unchanged if the
local references are connected.

Figure 4. Illegal connection between groups.

Figure 5. Circuit containing an illegal element connection. The numbers at the nodes indicate the
order in which terminals are discovered.

Figure 6. Collaboration diagram showing the violation detection algorithm.

Figure 7. A simple 2x2 spatial grid amplifier.

Figure 8. Local groups in a quasi-optical grid amplifier.

Figure 9. Transim netlist for the grid amplifier circuit of Figure 7.

Figure 6. Collaboration diagram showing the violation detection algorithm.

 Grid amplifier netlist
 *
 * Sweep in frequency
 .options f0 = \ 5.4e9 6.4e9 .1e9 \
 .hb fundamental = f0 spts = 6 oversample = 2
 ****************************************************
 * Definition of the ERA6 amplifier model
 * Input node = 1      Output node = 5    Common node = gnd1
 ****************************************************
 .subckt era6 1 5 "gnd1"
 * Input bias source.
 vsource:vindc 100 "gnd1" ampl=4.5
 res:rs1 100 1 r=50
 *
 * Output bias Source
 vsource:vcc 200 "gnd1" ampl=4.25
 res:rs2 200 4 r=50
 *
 vct:a11 2 54 3  gamma=-2.3 beta=8 kf=0.0025
 vct:b11 2 4 15  gamma=1.1 beta=1 kf=1
 + poly = <5.979287283058695e-4 -2.334033466242205e-02
 + 6.282215276145409e-02 -4.319540162756818e-02
 + 9.169340904889009e-03>
 ind:ssig11 54 "gnd1" l=1e-3
 cap:ssig11 54 4 c=1e-3
 ind:lin11 1 2 l=1.0238e-9
 cap:cin11 2 15 c= 0.32713e-12
 ymatrix:nin11 2 3 filename = "netin.yp"
 ind:le11 15 55 l=0.1e-9
 res:re11 55 "gnd1" r=7
 cap:ce11 55 "gnd1" c=1e-3
 ymatrix:ne11 3 15 filename = "freqcomp.yp"
 res:rout11 24 15 r=80
 cap:cdum11 24 4 c=1
 cap:cout11 4 15 c=0.64225e-12
 ind:lout11 4 5 l=1.2361e-9
 cap:coutsub11 5 "gnd1" c=0.12864e-12
 .ends
 ****************************************************
 * Identify local reference nodes
 ****************************************************
 .ref "LR1"
 .ref "LR2"   
 .ref "LR3"
 .ref "LR4"
 .ref "ind21"
 .ref "ind22"
 .ref "ind23"
 .ref "ind24"
 .ref "ind25"
 .ref "ind26"
 .ref "ind27"
 .ref "ind28"
 .ref "ind29"
 .ref "ind210"
 .ref "ind211"
 .ref "ind212"
 ****************************************************
 * Field Excitation
 ****************************************************
 gridex:iin 1   "LR1"  101 "LR1"
 +          5   "LR1"  105 "LR1"

 +          201 "LR2"  301 "LR2"
 +          205 "LR2"  305 "LR2"
 +          401 "LR3"  501 "LR3"
 +          405 "LR3"  505 "LR3"
 +          601 "LR4"  701 "LR4"
 +          605 "LR4"  705 "LR4"
 + "ind11" "ind21" "ind12" "ind22"
 + "ind13" "ind23" "ind14" "ind24"
 + "ind15" "ind25" "ind16" "ind26"
 + "ind17" "ind27" "ind18" "ind28"
 + "ind19" "ind29" "ind110" "ind210"
 + "ind111" "ind211" "ind112" "ind212"
 + ifilename = "fullgrid_cmp.i"
 + efilename = "horngrid.e" freq = f0
 *****************************************************
 * Ymatrix element Interface to Emag model
 *****************************************************
 ymatrix:ngrid 1 101 5 105 201 301 205 305
 + 401 501 405 505 601 701 605 705
 +"ind11" "ind12" "ind13" "ind14" "ind15" "ind16"
 +"ind17" "ind18"
 +"ind19" "ind110"  "ind111" "ind112"
 +"LR1" "LR2" "LR3" "LR4"
 +"ind21" "ind22" "ind23" "ind24" "ind25" "ind26"
 +"ind27" "ind28"
 +"ind29" "ind210"  "ind211" "ind212"
 + filename ="fullgrid_cmp.yp"
 ******************************************************
 *  Bias inductors
 ind:ind1 "ind11" "ind21"  l=11e-9 int_res=10.
 ind:ind2 "ind12" "ind22"  l=11e-9 int_res=10.
 ind:ind3 "ind13" "ind23"  l=11e-9 int_res=10.
 ind:ind4 "ind14" "ind24"  l=11e-9 int_res=10.
 ind:ind5 "ind15" "ind25"  l=11e-9 int_res=10.
 ind:ind6 "ind16" "ind26"  l=11e-9 int_res=10.
 ind:ind7 "ind17" "ind27"  l=11e-9 int_res=10.
 ind:ind8 "ind18" "ind28"  l=11e-9 int_res=10.
 ind:ind9 "ind19" "ind29"  l=11e-9 int_res=10.
 ind:ind10 "ind110" "ind210"  l=11e-9 int_res=10.
 ind:ind11 "ind111" "ind211"  l=11e-9 int_res=10.
 ind:ind12 "ind112" "ind212"  l=11e-9 int_res=10.
 *****************************
 * Unit cell 1 
 xamp11 1 5 "LR1" era6
 xamp12 101 105 "LR1" era6
 ******************************
 * Unit cell 2 
 xamp21 201 205 "LR2" era6
 xamp22 301 305 "LR2" era6 
 ******************************
 * Unit cell 3
 xamp31 401 405 "LR3" era6
 xamp32 501 505 "LR3" era6
 ******************************
 * Unit cell 4
 xamp41 601 605 "LR4" era6
 xamp42 701 705 "LR4" era6
 ******************************
 .end

Figure 9. Transim netlist for the grid amplifier circuit of Figure
7.


